
Abstract. We propose a mechanism for automating discovery of definitions,7

that, when added to a logic system for which we have a theorem prover, extends it8

to support an embedding of a new logic system into it. As a result, the synthesized9

definitions, when added to the prover, implement a prover for the new logic.10

As an instance of the proposed mechanism, we derive a Prolog theorem prover11

for an interesting but unconventional epistemic Logic by starting from the sequent12

calculus G4IP that we extend with operator definitions to obtain an embedding13

in intuitionistic propositional logic (IPC). With help of a candidate definition for-14

mula generator, we discover epistemic operators for which axioms and theorems15

of Artemov and Protopopescu’s Intuitionistic Epistemic Logic (IEL) hold and16

formulas expected to be non-theorems fail.17

We compare the embedding of IEL in IPC with a similarly discovered successful18

embedding of Dosen’s double negation modality, judged inadequate as an epis-19

temic operator. Finally, we discuss the failure of the necessitation rule for an20

otherwise successful S4 embedding and share our thoughts about the intuitions21

explaining these differences between epistemic and alethic modalities in the con-22

text of the Brouwer-Heyting-Kolmogorov semantics of intuitionistic reasoning23

and knowledge acquisition.24

Keywords: automatic synthesis of logic systems, deriving new theorem provers via25

program synthesis, epistemic intuitionistic logic, propositional intuitionistic logic, Prolog-26

based theorem provers, embedding of modal logics into intuitionistic logic.27

1 Introduction28

Deriving new logic systems and discovering relationships between them not only re-29

quires a knowledge-intensive understanding of the intricate connections between their30

axioms and inference rules but it is also a time-intensive trial and error process for the31

human logician. This is especially the case for logic systems that depart from the usual32

expectations coming from the prevalent use of classical logic in today’s computational33

tools and methodologies as well from familiarity with more commonly used forms of34

modal logic (e.g., alethic, temporal).35

This motivates our effort to explore ways to automate this process, resulting not36

only in discovering some salient relationships between new and well-established logic37



systems, but also in software artifacts (e.g., automated theorem provers) facilitating38

reasoning in these less explored new logics.39

Epistemic Logic Systems have been derived often in parallel and sometime as af-40

terthoughts of alethic Modal Logic Systems, in which modalities are defined by axioms41

and additional inference rules extending classical logic.42

In the context of Answer Set Programming (ASP) epistemic logics hosted in this43

framework like e.g., [1–3] show that intermediate logics1 (e.g., equilibrium logic, [4])44

can express epistemic operators by extending them with definitions of epistemic opera-45

tors.46

Steps2, further below classical logic or ASP, are taken in recent work [5], based47

on the Brouwer-Heyting-Kolmogorov (BHK) view of intuitionistic logic that takes into48

account the constructive nature of knowledge, modeling more accurately the connection49

between proof systems and the related mental processes. Along these lines, our inquiry50

into epistemic logic will focus on knowledge vs. truth seen as intuitionistic provability.51

Like in the case of embedding epistemic operators into ASP systems, but with au-52

tomation in mind, we will design a synthesis mechanism for epistemic operators via53

embedding in IPC. For this purpose we will generate candidate formulas that verify54

axioms, theorems and rules and fail on expected non-theorems. For this purpose, we55

will use a lightweight IPC theorem prover and we will also show that this view gener-56

alizes to a mechanism for discovering for a given modal logic, when possible, a simple57

embedding of the logic into IPC and derivation of a theorem prover for it.58

Our starting point is Artemov and Protopopescu’s Intuitionistic Epistemic Logic59

(IEL) [5] that will provide the axioms, theorems and non-theorems stating the require-60

ments that must hold for the definitions extending IPC. The discovery mechanism will61

also bring up Dosen’s interpretation of double negation [6] as a potential epistemic62

operator and we will look into applying the same discovery mechanisms to find an em-63

bedding of modal logic S4 in IPC, with special focus on the impact of the necessitation64

rule, which requires that all theorems of the logic are necessarily true.65

The rest of the paper is organized as follows. Section 2 overviews Artemov and66

Protopopescu’s Intuitionistic Epistemic Logic (IEL). Section 3 introduces the G4IP se-67

quent calculus prover for Intuitionistic Propositional Logic (IPC). Section 4 describes68

the generator for candidate formulas extending IPC with modal operator definitions.69

Section 5 explains the discovering of the definitions that ensure the embedding of IEL70

into IPC and the discovering of the embedding of Dosen’s double negation as a modal-71

ity operator. It also discusses the intuitions behind the embedding of IEL, including the72

epistemic equivalent of the necessity rule, in IPC and the adequacy of this embedding73

as a constructive mechanism for reasoning about knowledge. Section 6 studies the case74

of the S4 modal logic and the failure of the necessity rule, indicating the difficulty of75

embedding it in IPC by contrast to IEL. Section 7 overviews some related work and76

section 8 concludes the paper.77

1 Logics stronger than intuitionistic but weaker than classical.
2 Actually infinitely many, as there’s an infinite lattice of intermediate logics between classical

and intuitionistic logic.



The paper is written as a literate SWI-Prolog program with its extracted code at78

https://raw.githubusercontent.com/ptarau/TypesAndProofs/master/ieltp.pro.79

2 Overview of Artemov and Protopopescu’s IEL logic80

In [5] a system for Intuitionistic Epistemic Logic is introduced that81

“maintains the original Brouwer-Heyting-Kolmogorov semantics for intuition-82

ism and is consistent with the well-known approach that intuitionistic knowl-83

edge be regarded as the result of verification”.84

Instead of the classic, alethic-modalities inspired K operator for which85

KA→ A

Artemov and Protopopescu argue that co-reflection expresses better the idea of con-86

structivity of truth87

A→KA

They also argue that this applies to both belief and knowledge i.e., that88

“The verification-based approach allows that justifications more general than89

proof can be adequate for belief and knowledge”.90

On the other hand, they consider intuitionistic reflection acceptable, expressing the91

fact that “known propositions cannot be false”:92

KA→¬¬A

Thus, they position intuitionistic knowledge of A between A and ¬¬A and given93

that (via Glivenko’s transformation [7]) applying double negation to a formula embeds94

classical propositional calculus into IPC, they express this view as:95

Intuitionistic Truth ⇒ Intuitionistic Knowledge ⇒ Classical Truth.96

They axiomatize the system IEL as follows.97

98

1. Axioms of propositional intuitionistic logic;99

2. K(A→ B)→ (KA→KB); (distribution)100

3. A→KA. (co-reflection)101

4. KA→¬¬A (intuitionistic reflection)102

103

Rule Modus Ponens.104

They also argue that a weaker logic of belief (IEL−) is expressed by considering105

only axioms 1,2,3.106

3 The G4ip prover for IPC107

We will describe next our lightweight propositional intuitionistic theorem prover, that108

will be used to discover an embedding of IEL into IPC.109



3.1 The LJT/G4ip calculus, (restricted here to the implicational fragment)110

Motivated by problems related to loop avoidance in implementing Gentzen’s LJ calcu-111

lus, Roy Dyckhoff [8] introduces the following rules for the G4ip calculus3.112

113

LJT1 : A,Γ ` A114

115

116

LJT2 : A,Γ ` B
Γ ` A→B117

118

119

LJT3 : B,A,Γ ` G
A→B,A,Γ ` G120

121

122

LJT4 : D→B,Γ ` C→D B,Γ ` G
(C→D)→B,Γ ` G123

124

Note that LJT4 ensures termination as formulas in the sequent become smaller in a125

multiset ordering. The rules work with the context Γ being either a multiset or a set,126

and the calculus is sound and complete for IPC.127

128

For supporting negation, one also needs to add LJT5 that deals with the special term129

f alse. Then negation of A is defined as A→ f alse.130

131

132

LJT5 : f alse,Γ ` G133

134

135

Rules for conjunction, disjunction and bi-conditional (not shown here) are also part of136

the calculus.137

As it is not unusual with logic formalisms, the same calculus had been discovered138

independently in the 1950’s by Vorob’ev and in the 80’s-90’s by Hudelmaier [9, 10].139

3.2 A Lightweight Theorem Prover for Intuitionistic Propositional Logic140

Starting from the sequent calculus for the intuitionistic propositional logic in G4ip [8],141

to which we have also added rules for the “<->” relation, we obtain the following142

lightweight IPC prover.143

:- op(525, fy, ~ ).144

:- op(550, xfy, & ). % right associative145

:- op(575, xfy, v ). % right associative146

:- op(600, xfx, <-> ). % non associative147

prove_in_ipc(T):- prove_in_ipc(T,[]).148

149

3 Originally called the LJT calculus in [8]. Restricted here to its key implicational fragment.



prove_in_ipc(A,Vs):-memberchk(A,Vs),!.150

prove_in_ipc(_,Vs):-memberchk(false,Vs),!.151

prove_in_ipc(A<->B,Vs):-!,prove_in_ipc(B,[A|Vs]),prove_in_ipc(A,[B|Vs]).152

prove_in_ipc((A->B),Vs):-!,prove_in_ipc(B,[A|Vs]).153

prove_in_ipc(A & B,Vs):-!,prove_in_ipc(A,Vs),prove_in_ipc(B,Vs).154

prove_in_ipc(G,Vs1):- % atomic or disj or false155

select(Red,Vs1,Vs2),156

prove_in_ipc_reduce(Red,G,Vs2,Vs3),157

!,158

prove_in_ipc(G,Vs3).159

prove_in_ipc(A v B, Vs):-(prove_in_ipc(A,Vs);prove_in_ipc(B,Vs)),!.160

prove_in_ipc_reduce((A->B),_,Vs1,Vs2):-!,prove_in_ipc_imp(A,B,Vs1,Vs2).161

prove_in_ipc_reduce((A & B),_,Vs,[A,B|Vs]):-!.162

prove_in_ipc_reduce((A<->B),_,Vs,[(A->B),(B->A)|Vs]):-!.163

prove_in_ipc_reduce((A v B),G,Vs,[B|Vs]):-prove_in_ipc(G,[A|Vs]).164

prove_in_ipc_imp((C->D),B,Vs,[B|Vs]):-!,prove_in_ipc((C->D),[(D->B)|Vs]).165

prove_in_ipc_imp((C & D),B,Vs,[(C->(D->B))|Vs]):-!.166

prove_in_ipc_imp((C v D),B,Vs,[(C->B),(D->B)|Vs]):-!.167

prove_in_ipc_imp((C<->D),B,Vs,[((C->D)->((D->C)->B))|Vs]):-!.168

prove_in_ipc_imp(A,B,Vs,[B|Vs]):-memberchk(A,Vs).169

We validate it first by testing it on the implicational subset, derived via the Curry-170

Howard isomorphism [11], then against Roy Dyckhoff’s Prolog implementation4, work-171

ing on formulas up to size 12. Finally we run it on human-made tests5, on which we get172

no errors, solving correctly 161 problems, with a 60 seconds timeout, compared with173

the 175 problems solved by Roy Dyckhoff’s more refined, heuristics-based 400 lines174

prover, with the same timeout6. We refer to [11] for the derivation steps of variants175

of this prover working on the implicational and nested Horn clause fragments of IPC.176

While more sophisticated tableau-based provers are available for IPC among which we177

mention the excellent Prolog-based fCube [12], our prover’s compact size and adequate178

performance will suffice 7.179

4 The definition formula generator180

We start with a candidate formula generator that we will constrain further to be used for181

generating candidate definitions of our modal operators.182

4 https://github.com/ptarau/TypesAndProofs/blob/master/third_party/

dyckhoff_orig.pro
5 at http://iltp.de
6 https://github.com/ptarau/TypesAndProofs/blob/master/tester.pro
7 In fact, our prover is faster than both fCube and Dyckhoff’s prover on the set of formulas of

small size on which our definition induction algorithm will run.



4.1 Generating Operator Trees183

We generate all formulas of a given size by decreasing the available size parameter at184

each step when nodes are added to a tree representation of a formula. Prolog’s DCG185

mechanism is used to collect the leaves of the tree.186

genOperatorTree(N,Ops,Tree,Leaves):-187

genOperatorTree(Ops,Tree,N,0,Leaves,[]).188

189

genOperatorTree(_,V,N,N)-->[V].190

genOperatorTree(Ops,OpAB,SN1,N3)-->191

{ SN1>0,N1 is SN1-1,192

member(Op,Ops),make_oper2(Op,A,B,OpAB)193

},194

genOperatorTree(Ops,A,N1,N2),195

genOperatorTree(Ops,B,N2,N3).196

197

make_oper2(Op,A,B,OpAB):-functor(OpAB,Op,2),arg(1,OpAB,A),arg(2,OpAB,B).198

4.2 Synthesizing the definitions of modal operators199

As we design a generic definition discovery mechanism, we will denote generically our200

modal operators as follows.201

– “#” for “�”=necessary and “K”=known202

– “*” for “♦”=possible and “M”=knowable203

After the operator definitions204

:- op( 500, fy, #).205

:- op( 500, fy, *).206

we specify our generator as covering the usual binary operators and we constrain it to207

have at least one of the leaves of its generated trees to be a variable. Besides the false208

constant used in the definition of negation, we introduce also a new constant symbol “?”209

assumed not to occur in the language. Its role will be left unspecified until the possible210

synthesized definitions will be filtered. We will constrain candidate definitions to ensure211

that axioms and selected theorems hold and selected non-theorems fail.212

genDef(M,Def):-genDef(M,[(->),(&),(v)],[false,?],Def).213

214

genDef(M,Ops,Cs,(#(X):-T)):-215

between(0,M,N),216

genOperatorTree(N,Ops,T,Vs),217

pick_leaves(Vs,[X|Cs]),218

term_variables(Vs,[X]).219

Leaves of the generated trees will be picked from a given set.220

pick_leaves([],_).221

pick_leaves([V|Vs],Ls):-member(V,Ls),pick_leaves(Vs,Ls).222



We first expand our operator definitions for the “~” negation and “*” modal operator223

while keeping atomic variables and the special constant false untouched.224

expand_defs(_,false,R) :-!,R=false.225

expand_defs(_,A,R) :-atomic(A),!,R= A.226

expand_defs(D,~(A),(B->false)) :-!,expand_defs(D,A,B).227

expand_defs(D,*(A),R):-!,expand_defs(D,~ (# (~(A))),R).228

The special case for expanding a candidate operator definition D requires a fresh variable229

for each instance, ensured by Prolog’s built-in copy term.230

expand_defs(D,#(X),R) :-!,copy_term(D,(#(X):-T)),expand_defs(D,T,R).231

Other operators are traversed generically by using Prolog’s “=..” built-in and by re-232

cursing with expand def list on their arguments.233

expand_defs(D,A,B) :-234

A=..[F|Xs],235

expand_def_list(D,Xs,Ys),236

B=..[F|Ys].237

expand_def_list(_,[],[]).238

expand_def_list(D,[X|Xs],[Y|Ys]) :-239

expand_defs(D,X,Y),240

expand_def_list(D,Xs,Ys).241

The predicate prove with def refines our G4ip prover by first expanding the defini-242

tions extending IPC with a given candidate modality.243

prove_with_def(Def,T0) :-expand_defs(Def,T0,T1),prove_in_ipc(T1,[]).244

The definition synthesizer will filter the candidate definitions provided by genDef such245

that the predicate prove with def succeeds on all theorems and fails on all non-246

theorems, provided as names of the facts of arity 1 containing them.247

def_synth(M,D):-def_synth(M,iel_th,iel_nth,D).248

249

def_synth(M,Th,NTh,D):-250

genDef(M,D),251

forall(call(Th,T),prove_with_def(D,T)),252

forall(call(NTh,NT), \+prove_with_def(D,NT)).253

Note that the generator first builds smaller formulas and then larger ones up the specified254

maximum size.255

Example 1 Candidate definitions up to size 2256

?- forall(genDef(2,Def),println(Def)).257

#A :- A258

#A :- A -> A259

#A :- A -> false260

#A :- A -> ?261

#A :- false -> A262

#A :- ? -> A263



#A :- A & A264

#A :- A & false265

#A :- A & ?266

...267

#A :- (A -> ?) -> A268

...269

#A :- (? v A) v ?270

#A :- (? v false) v A271

#A :- (? v ?) v A272

5 Discovering the embedding of IEL and Dosen’s double negation273

modality in IPC274

We specify a given logic (e.g., IEL or S4) by stating theorems on which the prover275

extended with the synthetic definition should succeed and non-theorems on which it276

should fail.277

5.1 The discovery mechanism for IEL278

We start with the axioms of Artemov and Protopopescu’s IEL system:279

iel_th(a -> # a).280

iel_th(# (a->b)->(# a-> # b)).281

iel_th(# a -> ~ ~ a).282

Note that the axioms would be enough to specify the logic, but we also add some the-283

orems when intuitively relevant and/or mentioned in [5]. Our Prolog code, running in284

less than a second, is not slowed down by this in any significant way.285

iel_th(# (a & b) <-> (# a & # b)).286

iel_th(~ # false).287

iel_th(~ (# a & ~ a)).288

iel_th(~a -> ~ # a).289

iel_th( ~ ~ (# a -> a)).290

iel_th(# a & # (a->b) -> # b).291

iel_th(* (a & b) <-> (* a & * b)).292

iel_th(# a -> * a).293

iel_th(# a v # b -> # (a v b) ).294

iel_th(# p <-> # # p).295

iel_th(* a <-> * * a).296

iel_th(a -> *a).297

Again, following [5], we add our non-theorems.298

iel_nth(# a -> a).299

iel_nth(# (a v b) -> # a v # b).300

iel_nth(# a).301

iel_nth(~ (# a)).302

iel_nth(# false).303

iel_nth(# a).304



iel_nth(~ (# a)).305

iel_nth(* false).306

The necessitation rule in a modal logic requires that if T is a theorem than #T is also a307

theorem. This expresses the fact that the theorems of the logic are necessarily true, or308

in an epistemic context that if T is an (intuitionistically proven) theorem, then the agent309

knows T. Thus, we define (implicit) facts via a Prolog rule that states that the (generic)310

necessity operator “#” applied to proven theorems or axioms generates new theorems.311

iel_nec_th(T):-iel_th(T).312

iel_nec_th(# T):-iel_th(T).313

Finally, we obtain the discovery algorithm for IEL formula definitions and for IEL314

extended with the necessitation rule.315

iel_discover:-316

backtrack_over((def_synth(2,iel_th,iel_nth,D),println(D))).317

318

iel_nec_discover:-319

backtrack_over((def_synth(2,iel_nec_th,iel_nth,D),println(D))).320

321

backtrack_over(Goal):-call(Goal),fail;true.322

323

println(T):-numbervars(T,0,_),writeln(T).324

We run iel discover, ready to see the surviving definition candidates.325

Example 2 Definition discovery without the necessitation rule.326

?- iel_discover.327

#A:-(A->false)->A328

#A:-(A->false)->false329

#A:-(A-> ?)->A330

true.331

Example 3 Definition discovery with the necessitation rule.332

?- iel_nec_discover.333

#A:-(A->false)->A334

#A:-(A->false)->false335

#A:-(A-> ?)->A336

true.337

Unsurprisingly, the results are the same, as a consequence of axiom A -> #A.338

Clearly, the formula #A:-(A->false)->A is not interesting as it would define339

knowing something as a contradiction that implies itself.340

This brings us to the second definition formula candidate.341

5.2 Eliminating Dosen’s double negation modality342

In [2] double negation in IPC is interpreted as a “�” modality. This corresponds to one343

of the synthetic definitions #A :- (A->false)->false that is equivalent in IPC to344



#A :- ~~A. It is argued in [5] that it does not make sense as an epistemic modality,345

mostly because it would entail that all classical theorems are known intuitionistically.346

We eliminate it by requiring the collapsing of “*” into “#” to be a non-theorem:347

iel_nth(* a <-> # a).348

In fact, while known (#) implies knowable (~#~ = *), it is reasonable to think, as in349

most modal logics, that the inverse implication does not hold.350

After that, we have:351

Example 4 The double negation modality is eliminated, as it collapses # and *.352

?- iel_discover.353

#A:-(A -> ?)->A354

true.355

356

?- iel_nec_discover.357

#A:-(A -> ?)->A358

true.359

5.3 Knowledge as awareness?360

This leaves us with the #A :- (A -> ?) -> A.361

Among the consequences of the fact that intuitionistic provability strictly implies362

classical, is that there’s plenty of room left between p and ~~p, where both # and * find363

their place, given that the following implication chain holds.364

p -> #p -> *p -> ~~p365

Let us now find an (arguably) intuitive meaning for the “?” constant in the definition.366

The interpretation of knowledge as awareness about truth goes back to [13]. Our final367

definition of intuitionistic epistemic modality as “#A :- (A -> ?) -> A” suggests368

interpreting “?” as awareness of an agent entailed by (a proof of) A. With this in mind,369

one obtains an embedding of IEL in IPC via the extension370

KA ≡ (A→ eureka)→ A

where eureka is a new symbol not occurring in the language8.371

In line with the Brouwer-Heyting-Kolmogorov (BHK) interpretation of intuitionis-372

tic proof, we may say that an agent knows A if and only if A is validated by a proof of A373

that induces awareness of the agent about it.374

Thus knowledge of an agent, in this sense, collects facts that are proven construc-375

tively in a way that is “understood” by the agent. The consequence376

KA→¬¬A

would then simply say that intuitionistic truths, that the agent is aware of, are also377

classically valid.378

Thus, we can define our newly synthesized prover for IEL as follows.379

8 Not totally accidentally named, given the way Archimedes expressed his sudden awareness
about the volume of water displaced by his immersed body.



iel_prove(P):-prove_with_def((#A :- (A -> eureka) -> A),P).380

Interestingly, if one allows eureka to occur in the formulas of the language given as381

input to the prover, then it becomes (the unique) value for which we have equivalence382

between being known and having a proof.383

?- iel_prove(#eureka <-> eureka).384

true .385

Similarly, it would also follow that386

?- iel_prove(*eureka <-> ~ ~ eureka).387

true.388

Thus, one would need to forbid accepting it as part of the prover’s language to closely389

follow the intended semantics of IEL.390

5.4 Discussion391

The most significant consequence of the successful embedding of IEL into IPC via the392

epistemic modality definition #A :- (A -> eureka) -> A) is that we have actually393

derived a theorem prover for IEL. The theorem prover is implemented by the predicate394

iel prove/1 by extending a theorem prover for IPC with the induced definition.395

As the IPC fragment with two variables, implication and negation has exactly 518396

equivalence classes of formulas [14, 15], one would expect the construction deriving397

“*” from “#” to reach a fixpoint. We can use our prover to find out when that happens.398

?- iel_prove(#p <-> ~ # (~p)).399

false.400

iel_prove(*p <-> ~(*(~p))).401

true.402

Thus the fixpoint of the construction is “*”, that we have interpreted as meaning that a403

proposition is knowable. Therefore, the equivalence reads reasonably as “something is404

knowable if and only if its negation is not knowable”. Note also that405

?- iel_prove(~(*(~p)) -> #p).406

false.407

fails, by contrast to the equivalence �p≡ ¬♦¬p usual in classical modal logics.408

6 Discovering an embedding of S4 without the necessitation rule409

The fact that both IPC and S4 are known to be PSPACE-complete [16] means that410

polynomial-time translations exist between them.411

In fact, Gödel’s translation from IPC to S4 (by prefixing each subformula with the412

� operator) shows that the embedding of IPC into S4 can be achieved quite easily, by413

using purely syntactic means. However, the (very) few papers attempting the inverse414

translation [17, 18] rely on methods often involving intricate semantic constructions.415

We will use our definition generator to identify the problem that precludes a simple416

embedding of S4 into IPC.417

We start with the axioms of S4.418



s4_th(# a -> a).419

s4_th(# (a->b) -> (# a -> # b)).420

s4_th(# a -> # # a).421

We add a few theorems.422

s4_th(* * a <-> * a).423

s4_th(a -> * a).424

s4_th(# a -> * a).425

s4_th(# a v # b -> # (a v b)).426

s4_th(# (a v b) -> # a v # b).427

We add some non-theorems that ensure additional filtering.428

s4_nth(# a).429

s4_nth(~ (# a)).430

s4_nth(# false).431

s4_nth(* false).432

s4_nth(* a -> # * a). % true only in S5433

s4_nth(a -> # a).434

s4_nth(* a -> a).435

s4_nth(# a <-> ?).436

s4_nth(* a <-> ?).437

Like in the case of IEL we define implicit facts stating that the necessitation rule holds.438

s4_nec_th(T):-s4_th(T).439

s4_nec_th(# T):-s4_th(T).440

Finally we implement the definition discovery predicates and run them.441

s4_discover:-442

backtrack_over((def_synth(2,s4_th,s4_nth,D),println(D))).443

444

s4_nec_discover:-445

backtrack_over((def_synth(2,s4_nec_th,s4_nth,D),println(D))).446

Example 5 The necessitation rule eliminates all simple embeddings of S4 into IPC,447

while a lot of definition formulas pass without it.448

?- s4_discover.449

#A :- A & ?450

#A :- ? & A451

#A :- A & (A-> ?)452

#A :- A & (? -> false)453

...454

true.455

456

?- s4_nec_discover.457

true.458

Among the definitions succeeding without passing the necessity rule test, one might459

want to pick #A :- ? & A as an approximation of the S4 “�” operator. In this case “?”460



would simply state that “the IPC prover is sound and complete”. Still, given the failure461

of the necessitation rule, the resulting logic is missing a key aspect of the intended462

meaning of S4-provability.463

7 Related work464

Program synthesis techniques have been around in logic programming with the advent465

of Inductive Logic Programming [19], but the idea of learning Prolog programs from466

positive and negative examples goes back to [20]. Our definition synthesizer fits in this467

paradigm, with focus on the use of a theorem prover of a decidable logic (IPC) filtering468

formulas provided by a definition generator through theorems as positive examples and469

non-theorems as negative examples. The means we use for our definition synthesis are470

in fact as simple as those described in [20]. The strength of our approach comes from471

the use of a theorem prover that efficiently validates or rejects definition candidates.472

The idea to use the new constant “?” in our synthesizer is inspired by proofs that some473

fragments of IPC reduced to two variables have a (small) finite number of equivalence474

classes [14, 15] as well as by the introduction of new variables, in work on polynomial475

embeddings of S4 into IPC [17, 18].476

We refer to [5] for a thorough discussion of the merits of IEL compared to epis-477

temic logics following closely classical modal logic, but the central idea about using478

intuitionistic logic is that of belief and knowledge as the product of verification. Our479

embedding of IEL in IPC can be seen as a simplified view of this process through a480

generic “awareness of an agent” concept in line with [13].481

In [1] the concept of epistemic specifications is introduced that support express-482

ing knowledge and belief in an Answer Set Programming framework. Interestingly,483

refinements of this work like [21] and [3] discuss difficulties related to expressing an484

assumption like p→Kp in terms of ASP-based epistemic operators.485

Equilibrium logic [4] gives a semantics to Answer Set programs by extending the486

3-valued intermediate logic of here-and-there HT with Nelson’s constructive strong487

negation. In [22] a 5-valued truth-table semantics for equilibrium logic is given. In [23]488

(and several other papers) epistemic extensions of equilibrium logic [4] are proposed, in489

which Kp→ p. By contrast to “alethic inspired” epistemic logics postulating Kp→ p490

we closely follow the p→Kp view on which [5] is centered.491

While we have eliminated Dosen’s double negation modality [6] as an epistemic492

operator Kp ≡ ¬¬p, it is significant that it came out as the only other meaningful493

candidate produced by our definition synthesizer.494

This suggests that it might be worth investigating further how a similar definition495

discovery mechanism as the one we have used for IEL and S4 would work for logics496

with multiple negation operators like equilibrium logic.497

Besides the Kp→ p vs. p→ Kp problem a more general question is the choice498

of the logic supporting the epistemic operators, among logics with finite truth-value499

models (e.g., classical logic or equilibrium logic) or, at the limit, intuitionistic logic500

itself, with no such models. Arguably, this could be application dependent, as epistemic501

operators built on top of IPC are likely to fit better the landscape with intricate nuances502

of a richer set of epistemic and doxastic operators, while such operators built on top of503



finite-valued intermediate logics would benefit from simpler decision procedures and504

faster evaluation mechanisms.505

8 Conclusions506

We have devised a general mechanism for synthesizing definitions that extend a given507

logic system endowed with a theorem prover. The set of theorems on which the ex-508

tended prover should succeed and the set of non-theorems on which it should fail can509

be seen as a declarative specification of the extended system. Success of the approach510

on embedding the IEL system in IPC and failure on trying to embed S4 has revealed511

the individual role of the axioms, theorems and rules that specify a given logic system512

and their interaction with the necessitation rule .513

Given its generality, our definition generation technique can be applied also to epis-514

temic or modal logic axiom systems to find out if they have interesting embeddings in515

ASP and superintuitionistic logics for which high quality solvers or theorem provers516

exist. Our program synthesis process, when the embedding succeeds, provides a way to517

automate the exploration of a new logic system with help of its derived theorem prover518

and facilitates the work of the human logician to validate or invalidate the intuitions519

behind it.520

